

# NAG Library Routine Document

## F07TUF (ZTRCON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07TUF (ZTRCON) estimates the condition number of a complex triangular matrix.

### 2 Specification

```
SUBROUTINE F07TUF (NORM, UPLO, DIAG, N, A, LDA, RCOND, WORK, RWORK,
                  INFO)
INTEGER          N, LDA, INFO
REAL (KIND=nag_wp) RCOND, RWORK(N)
COMPLEX (KIND=nag_wp) A(LDA,*), WORK(2*N)
CHARACTER(1)    NORM, UPLO, DIAG
```

The routine may be called by its LAPACK name *ztrcon*.

### 3 Description

F07TUF (ZTRCON) estimates the condition number of a complex triangular matrix  $A$ , in either the 1-norm or the  $\infty$ -norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that  $\kappa_\infty(A) = \kappa_1(A^T)$ .

Because the condition number is infinite if  $A$  is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine computes  $\|A\|_1$  or  $\|A\|_\infty$  exactly, and uses Higham's implementation of Hager's method (see Higham (1988)) to estimate  $\|A^{-1}\|_1$  or  $\|A^{-1}\|_\infty$ .

### 4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

### 5 Arguments

- 1: NORM – CHARACTER(1) *Input*  
*On entry:* indicates whether  $\kappa_1(A)$  or  $\kappa_\infty(A)$  is estimated.  
 NORM = '1' or 'O'  
 $\kappa_1(A)$  is estimated.  
 NORM = 'I'  
 $\kappa_\infty(A)$  is estimated.  
*Constraint:* NORM = '1', 'O' or 'I'.

- 2: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies whether  $A$  is upper or lower triangular.  
 UPLO = 'U'  
      $A$  is upper triangular.  
 UPLO = 'L'  
      $A$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 3: DIAG – CHARACTER(1) *Input*  
*On entry:* indicates whether  $A$  is a nonunit or unit triangular matrix.  
 DIAG = 'N'  
      $A$  is a nonunit triangular matrix.  
 DIAG = 'U'  
      $A$  is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.  
*Constraint:* DIAG = 'N' or 'U'.
- 4: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 5: A(LDA,\*) – COMPLEX (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  triangular matrix  $A$ .  
     If UPLO = 'U',  $A$  is upper triangular and the elements of the array below the diagonal are not referenced.  
     If UPLO = 'L',  $A$  is lower triangular and the elements of the array above the diagonal are not referenced.  
     If DIAG = 'U', the diagonal elements of  $A$  are assumed to be 1, and are not referenced.
- 6: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07TUF (ZTRCON) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 7: RCOND – REAL (KIND=nag\_wp) *Output*  
*On exit:* an estimate of the reciprocal of the condition number of  $A$ . RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**,  $A$  is singular to working precision.
- 8: WORK(2 × N) – COMPLEX (KIND=nag\_wp) array *Workspace*
- 9: RWORK(N) – REAL (KIND=nag\_wp) array *Workspace*
- 10: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value  $\rho$ , and in practice is nearly always less than  $10\rho$ , although examples can be constructed where RCOND is much larger.

## 8 Parallelism and Performance

F07TUF (ZTRCON) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

A call to F07TUF (ZTRCON) involves solving a number of systems of linear equations of the form  $Ax = b$  or  $A^Hx = b$ ; the number is usually 5 and never more than 11. Each solution involves approximately  $4n^2$  real floating-point operations but takes considerably longer than a call to F07TSF (ZTRTRS) with one right-hand side, because extra care is taken to avoid overflow when  $A$  is approximately singular.

The real analogue of this routine is F07TGF (DTRCON).

## 10 Example

This example estimates the condition number in the 1-norm of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.78 + 4.56i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.00 - 0.30i & -4.11 + 1.25i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.89 - 1.34i & 2.36 - 4.25i & 4.15 + 0.80i & 0.00 + 0.00i \\ -1.89 + 1.15i & 0.04 - 3.69i & -0.02 + 0.46i & 0.33 - 0.26i \end{pmatrix}.$$

The true condition number in the 1-norm is 70.27.

### 10.1 Program Text

```

Program f07tufe

!      F07TUF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, ztrcon
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: diag = 'N', norm = '1'
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: rcond
      Integer                     :: i, info, lda, n
      Character (1)               :: uplo
!      .. Local Arrays ..

```

```

      Complex (Kind=nag_wp), Allocatable :: a(:,,:), work(:)
      Real (Kind=nag_wp), Allocatable :: rwork(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07TUF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),work(2*n),rwork(n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If

!      Estimate condition number
!      The NAG name equivalent of ztrcon is f07tuf
      Call ztrcon(norm,uplo,diag,n,a,lda,rcond,work,rwork,info)

      Write (nout,*)
      If (rcond>=x02ajf()) Then
        Write (nout,99999) 'Estimate of condition number =', 1.0_nag_wp/rcond
      Else
        Write (nout,*) 'A is singular to working precision'
      End If

99999 Format (1X,A,1P,E10.2)
      End Program f07tufe

```

## 10.2 Program Data

F07TUF Example Program Data

```

  4                                     :Value of N
  'L'                                   :Value of UPLO
( 4.78, 4.56)
( 2.00,-0.30) (-4.11, 1.25)
( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
(-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26) :End of matrix A

```

## 10.3 Program Results

F07TUF Example Program Results

Estimate of condition number = 3.74E+01

---